Solutions to Problem Set 5, PHSY 350, Spring 2009

Problem 1

Preparation

In this problem we are asked to plot the motion of a damped harmonic oscillator nine different times: we are given three setes of initial conditions and for each set of initial conditions we are to plot overdamped, critically damped, and underdamped motion. We can make this a lot easier on ourselves by defining several functions right at the beginning in a way that allows us to write as little repeated code as possible.

Because the three different solutions (over, critical and underdamping) have three different mathematical forms it is easiest to group our function definitions into these three categories.

All three cases have the same frequency ω so it makes sense to define that now,

```
> omega:=1:
```

Overdamping

In this case Boas says the solution is

$$y(t) = A\,e^{-\lambda t} + Be^{-\mu t}$$

where λ and μ are related to the natural frequency ω and damping b by

$$\lambda = b + \sqrt{b^2 - \omega^2} \quad \text{and} \quad \mu = b - \sqrt{b^2 - \omega^2}.$$

For overdamping we are told to use $b = \frac{13}{5}$, so that

```
> b:=13/5: lambda:= b+sqrt(b^2-omega^2);
```

$$\lambda := 5 \qquad \textbf{(1.1.1.1)}$$

```
> mu:=b-sqrt(b^2-omega^2);
```

$$\mu := \frac{1}{5} \qquad \textbf{(1.1.1.2)}$$

We can write the overdamped solution as

```
> y_over:=(t,y0,v0)->A_over(y0,v0)*exp(-5*t)+B_over(y0,v0)*
  exp(-t/5):
```

and the constants A and B are determined by initial conditions and will be different for each part. The initial position is

$$y(0) = A + B$$

and the velocity is $v(t) = \frac{dy}{dt} = -5\,Ae^{-5\,t} - \frac{1}{5}\,Be^{-\frac{t}{5}}$, so that the initial velocity is

$v(0) = -5A - \frac{B}{5}$.

If we make Maple figure out A and B for us,

```
> eq1:=y0=A+B: eq2:=v0=-(5*A+B/5):
> solve([eq1,eq2],[A,B]);
```

$$\left[\left[A = -\frac{1}{24}\,y0 - \frac{5}{24}\,v0,\ B = \frac{5}{24}\,v0 + \frac{25}{24}\,y0\right]\right] \qquad (1.1.1.3)$$

then we can define functions for A and B too,

```
> A_over:=(y0,v0)->-y0/24-5*v0/24:
> B_over:=(y0,v0)->5*v0/24+25*y0/24:
```

Critical damping

The solution in the critically damped case, which occurs when $\omega = b$, is

$y(t) = (A + Bt)e^{-\omega t}$,

which in Maple is

```
> y_crit:=(t,y0,v0)->(A_crit(y0,v0)+B_crit(y0,v0)*t)*exp(-
  omega*t):
```

In this case the intial conditions are related to A and B by

$y(0) = A$ and $v(t) = Be^{-\omega t} - \omega(A + Bt)e^{-\omega t}$ so that $v(0) = B - \omega A$. Solving this for B gives $B = v(0) + \omega y(0)$.

```
> A_crit:=(y0,v0)->y0:
> B_crit:=(y0,v0)->v0+omega*y0:
```

Underdamping

The solution in this case is

$y(t) = e^{-bt}\left(A\cos\left(\omega_1 t\right) + B\sin\left(\omega_1 t\right)\right)$

where $\omega_1 = \sqrt{\omega^2 - b^2}$, and Boas says we should use $b = \frac{5}{13}$ for this case, so that

```
> b:=5/13: omega1:=sqrt(omega^2-b^2);
```

$$\omega 1 := \frac{12}{13} \qquad (1.1.3.1)$$

and the function $y(t)$ is

```
> y_under:=(t,y0,v0)->exp(-b*t)*(A_under(y0,v0)*cos(omega1*t)
  +B_under(y0,v0)*sin(omega1*t)):
```

The constants A and B are related to the initial positions. The intiial position is $y(0) = A$. The velocity is

$v(t) = -b\,e^{-bt}\left(A\cos\left(\omega_1 t\right) + B\sin\left(\omega_1 t\right)\right) + e^{-bt}\omega_1\left(-A\sin\left(\omega_1 t\right) + B\cos\left(\omega_1 t\right)\right)$,

so that $v(0) = -b\,A + \omega_1 B$. It seems a little silly to define functions for A and B in this case but I will to be consistent with the other two cases.

```
> A_under:=(y0,v0)->y0:
> B_under:=(y0,v0)->(1/omega1)*(v0+b*y0):
```

The Plots

To make this a bit easier to read I will write out each of the functions we are plotting, then stick those into plot itself. I'm not sure that sentence made sense, but I assure you it will once you've read the bit of code below. Of course, that assumes that someone will actually read the sentence. I really hope so because it is late, I am tired, and I could have gone to bed over an hour ago if I had known no one would read this riveting prose.

But I digress.

Part a, y(0)=1, v(0)=0

```
> y0:=1: v0:=0:
> over:=y_over(t,y0,v0); crit:=y_crit(t,y0,v0); under:=
  y_under(t,y0,v0);
```

$$over := -\frac{1}{24}\,e^{-5\,t} + \frac{25}{24}\,e^{-\frac{1}{5}\,t}$$

$$crit := (1+t)\,e^{-t}$$

$$under := e^{-\frac{5}{13}\,t}\left(\cos\left(\frac{12}{13}\,t\right) + \frac{5}{12}\sin\left(\frac{12}{13}\,t\right)\right) \qquad \textbf{(1.2.1.1)}$$

```
> plot([over,crit,under],t=0..20,linestyle=[1,2,3],color=
  black,legend=["Overdamped","Critical damping",
  "Underdamped"],title="y(0)=1, v(0)=0");
```



Part b, y(0)=1, v(0)=1

```
> y0:=1: v0:=1:
> over:=y_over(t,y0,v0); crit:=y_crit(t,y0,v0); under:=
  y_under(t,y0,v0);
```

$$over := -\frac{1}{4}\,e^{-5t} + \frac{5}{4}\,e^{-\frac{1}{5}t}$$

$$crit := (1+2t)\,e^{-t}$$

$$under := e^{-\frac{5}{13}t}\left(\cos\left(\frac{12}{13}t\right) + \frac{3}{2}\sin\left(\frac{12}{13}t\right)\right) \tag{1.2.2.1}$$

```
> plot([over,crit,under],t=0..20,linestyle=[1,2,3],color=
  black,legend=["Overdamped","Critical damping",
  "Underdamped"],title="y(0)=1, v(0)=1");
```



Part c, y(0)=1, v(0)=-1

```
> y0:=1: v0:=-1:
> over:=y_over(t,y0,v0); crit:=y_crit(t,y0,v0); under:=
  y_under(t,y0,v0);
```

$$over := \frac{1}{6}\,e^{-5t} + \frac{5}{6}\,e^{-\frac{1}{5}t}$$

$$crit := e^{-t}$$

$$under := e^{-\frac{5}{13}t}\left(\cos\left(\frac{12}{13}t\right) - \frac{2}{3}\sin\left(\frac{12}{13}t\right)\right) \quad \textbf{(1.2.3.1)}$$

```
> plot([over,crit,under],t=0..20,linestyle=[1,2,3],color=
black,legend=["Overdamped","Critical damping",
"Underdamped"],title="y(0)=1, v(0)=1");
```



Note that in all three cases the critical damping solution is the one that approaches zero most rapidly.

Problem 2

```
> restart:  with(plots):
```

Definition of cosine and sine fourier series

From the Supplement to Lab 6: cosine series routines

Since I will be needed to compute Fourier cosine series later on in this lab activity, I am going to copy those functions from the lab supplement to the begining of this worksheet.

```
> assume(n,integer);assume(m,integer):
```

```
> c := (x,n) -> cos((2*Pi*n*x)/L):
> A:=proc(expr,var,n)
          simplify(int(expr*c(var,n),var=0..L)/int(c(var,n)*c
  (var,n),var=0..L));
  end proc:
> cosineFS:=proc(expr,var,n)
          A(expr,var,0)+sum(A(expr,var,m)*c(var,m),m=1..n);
  end proc:
> cosineFP:=proc(expr,var,n)
          A(expr,var,0)+add(A(expr,var,m)*c(var,m),m=1..n);
  end proc:
```

From the solutions to Lab 6: sine series rountines

a. We are asked to develop routines for computing the Fourier sine series. These routines are very similar to the Fourier cosine series routines from the supplement, but tweaked in the appropriate fashion (notice there is no n=0 term in the Fourier sine series, or rather it goes to 0.):

```
> s := (x,n) -> sin((2*Pi*n*x)/L):
> B:=proc(expr,var,n)
          simplify(int(expr*s(var,n),var=0..L)/int(s(var,n)*s
  (var,n),var=0..L));
  end proc:
> sineFS:=proc(expr,var,n)
          sum(B(expr,var,m)*s(var,m),m=1..n);
  end proc:
> sineFP:=proc(expr,var,n)
          add(B(expr,var,m)*s(var,m),m=1..n);
  end proc:
```

Error in the sine series approximation

The function we are to analyze is

```
> f := x-> x-L/2:
```

We first estimate the error if we end the sine series at n=15 by calcuating B_{n+1}

```
> L:=1: abs(B(f(x),x,16))=evalf(abs(B(f(x),x,16)));
```

$$\frac{1}{16\,\pi} = 0.01989436788 \qquad \textbf{(2.2.1)}$$

Next we are to calculate the actual error. We begin by defining a function that is the sine series expansion of $f(x)$ to 15th order,

```
> fsine15 := unapply(sineFP(f(x),x,15),x):
```

The error we want is the average value of $(f\text{-}fsine15)^2$ over the interval from 0 to L. That average, which I will call *Erravg* is most esily calculated with an integral,

$$Erravg = \frac{1}{L}\int_0^L (f\text{-}fsine15)^2\,dx.$$

```
> Erravg:= 1/L*int((f(x)-fsine15(x))^2,x=0..L);evalf(Erravg);
```

$$Erravg := \frac{1}{259718659200}\,\frac{-205234915681+21643221600\,\pi^2}{\pi^2}$$

$$0.003267293245 \qquad \textbf{(2.2.2)}$$

```
>
```

This seems a lot smaller than the error we estimated in (1.2.1)...what is going on? We have found the average *square* difference between the function and its approximation, so let's take the square root and see what we get

```
> evalf(sqrt(Erravg));
```

$$0.05716024182 \quad (2.2.3)$$

This seems closer, though now it seems we underestimated the error. Just for fun, here is a plot of the function and its approximation.

```
> plot([f(x),fsine15(x)],x=0..L,legend=["f(x)","approx"]);
```



3 (a) We are given that $h(x) = h_s(x) + h_a(x)$, where $h_a(x)$ is antisymmetric and $h_s(x)$ is symmetric. Then $h(-x) = h_s(-x) + h_a(-x) = h_s(x) - h_a(x)$, since $h_s(-x) = h_s(x)$ and $h_a(-x) = -h(x)$.

(b) If we add $h(x)$ and $h(-x)$ we get

$$h(x) + h(-x) = (h_s(x) + h_a(x)) + (h_s(x) - h_a(x)) = 2h_s(x)$$
$$\Rightarrow \quad h_s(x) = \frac{h(x) + h(-x)}{2}. \quad (1)$$

Similarly, if we subtract $h(-x)$ from $h(x)$ we obtain

$$h(x) - h(-x) = (h_s(x) + h_a(x)) - (h_s(x) - h_a(x)) = 2h_a(x)$$
$$\Rightarrow \quad h_a(x) = \frac{h(x) - h(-x)}{2}. \quad (2)$$

(c) For the function $h(x) = (x^3 + x^2)\sin(x)$ we have $h(-x) = (-x^3 + x^2)(-\sin x) = x^3 \sin x - x^2 \sin x$. Using Eq. (1) we have

$$h_s(x) = \frac{1}{2}(h(x) + h(-x)) = x^3 \sin x, \quad (3)$$

and for the odd part

$$h_a(x) = \frac{1}{2}(h(x) - h(-x)) = x^2 \sin x. \quad (4)$$

4 We are to change variables in the four functions in Lab 6 so that we can evaluate whether the functions are symmetric, antisymmetric, or neither. The change of variables is $\tilde{x} = x - L/2$ so that $x = \tilde{x} + L/2$. For each of the functions we have

$$x(L - x) = (\tilde{x} + L/2)(L - (\tilde{x} + L/2)) = -\left(\tilde{x}^2 - L^2/4\right) \quad (5a)$$
$$(x - L/2)^3 = (\tilde{x} + L/2 - L/2)^3 = \tilde{x}^3 \quad (5b)$$
$$|x - L/2| = |\tilde{x} + L/2 - L/2| = |\tilde{x}| \quad (5c)$$
$$f(\tilde{x}) = \begin{cases} -1, & \tilde{x} < 0 \\ 1, & \tilde{x} > 0. \end{cases} \quad (5d)$$

Written in this form it is clear that the first and third functions are symmetric (and so can be written with cosine series only) and the second and fourth functions are odd (and so can be written with a sine series only).